AFFINE SPACE

(Exercise 7, sheet 4) In the affine space (A_3, V_3, f) we consider the coordinates systems $R=\{O;B_V=\{e_1, e_2, e_3\}\}$ and $R'=\{O'; B_V'=\{v_1, v_2, v_3\}\}$ where O' with respect to R has coordinates (-1,0,0) and v_1, v_2, v_3 are vectors with coordinates (1,1,0), (0,-1,0) and (0,0,-1) respectively, in B_V.

a) Find the equations of the change of the coordinates from *R*' to *R* and from *R* to *R*'. With respect to *R*' find the coordinates of *P* whose coordinates are (1,2,-1) with respect of the coordinates system *R*

b) Find with respect of the coordinate system *R*' the parametric and cartesian equations of the plane π knowing that with respect of *R* it has cartesian equation $2x_1 - x_2 + x_3 + 2 = 0$.

c) Find with respect of the coordinate system *R*′ the continuous form and the cartesian equations of the line *r* that has the following equations with respect to *R*

$$2x_1 + x_2 = 0,$$

$$x_1 - 2x_2 + x_3 = 1.$$

Solution

The matrix of change of coordinates from *R*' to *R* is determined by the coordinates of O' with respect to *R* and the coordinates of the vectors of the basis *B*' in the basis *B*.

```
> restart;
> with(linalg):
```

a)

```
> Op:=[-1,0,0];
  v1:=[1,1,0];
  v2:=[0,-1,0];
  v3:=[0,0,-1];
```

$$Op := [-1, 0, 0]$$
$$v1 := [1, 1, 0]$$
$$v2 := [0, -1, 0]$$
$$v3 := [0, 0, -1] \quad \textbf{(1.1.1.1)}$$

```
> MRpR:=stackmatrix([1,0,0,0],concat(Op,v1,v2,v3));
```

(1.1.1.2)

$$MRpR := \begin{bmatrix} 1 & 0 & 0 & 0 \\ -1 & 1 & 0 & 0 \\ 0 & 1 & -1 & 0 \\ 0 & 0 & 0 & -1 \end{bmatrix} \quad \textbf{(1.1.1.2)}$$

```
> MRRp:=inverse(MRpR);
```

$$MRRp := \begin{bmatrix} 1 & 0 & 0 & 0 \\ 1 & 1 & 0 & 0 \\ 1 & 1 & -1 & 0 \\ 0 & 0 & 0 & -1 \end{bmatrix} \quad \textbf{(1.1.1.3)}$$

The expressions of the changes of coordinates are:

```
> evalm(MRpR&*[1,y1,y2,y3]);
```

$$\begin{bmatrix} 1 & -1+y1 & y1-y2 & -y3 \end{bmatrix} \quad \textbf{(1.1.1.4)}$$

```
> RpR:=(y1,y2,y3)->[-1+y1, y1-y2, -y3];
```

$$RpR := (y1, y2, y3) \rightarrow [-1+y1, y1-y2, -y3] \quad \textbf{(1.1.1.5)}$$

```
> evalm(MRRp&*[1,x1,x2,x3]);
```

$$\begin{bmatrix} 1 & 1+x1 & 1+x1-x2 & -x3 \end{bmatrix} \quad \textbf{(1.1.1.6)}$$

```
> RRp:=(x1,x2,x3)->[1+x1, 1+x1-x2, -x3];
```

$$RRp := (x1, x2, x3) \rightarrow [1+x1, 1+x1-x2, -x3] \quad \textbf{(1.1.1.7)}$$

The coordinates of P in R' are:

```
> RRp(1,2,-1);
```

$$[2, 0, 1] \quad \textbf{(1.1.1.8)}$$

b)

```
> pi:=2*x1-x2+x3+2=0;
```

$$\pi := 2\,x1 - x2 + x3 + 2 = 0 \quad \textbf{(1.1.2.1)}$$

```
> solve(2*x1-x2+x3+2=0);
```

$$\{x1 = x1, x2 = 2\,x1 + x3 + 2, x3 = x3\} \quad \textbf{(1.1.2.2)}$$

```
> par_piR:=[a,2+2*a+b,b];
```

$$par\_piR := [a, 2 + 2\,a + b, b] \quad \textbf{(1.1.2.3)}$$

```
> P:=[-1,0,0];
  v1:=[1,2,0];
  v2:=[0,1,1];
```

$$P := [-1, 0, 0]$$
$$v1 := [1, 2, 0]$$
$$v2 := [0, 1, 1] \quad \textbf{(1.1.2.4)}$$

The direction of the plane (or vector subspace associated) is generatec by v1 and v2, <v1,v2>.
The parametric equations of the plane in R' are:

```
> RRp(a, 2+2*a+b, b);
```

$$[1 + a, -1 - a - b, -b] \quad \textbf{(1.1.2.5)}$$

The coordiantes of a point Q of Pi in R' and the coordiantes of two vectors w1 and w2of the direcction of Pi in the basis B' are:

```
> QRp:=[1,-1,0];w1Bp:=[1,-1,0];w2Bp:=[0,-1,-1];
```

$$QRp := [1, -1, 0]$$
$$w1Bp := [1, -1, 0]$$
$$w2Bp := [0, -1, -1] \quad \textbf{(1.1.2.6)}$$

```
> Y:=[y1,y2,y3]:
> N:=matrix([Y-QRp,w1Bp,w2Bp]);
```

$$N := \begin{bmatrix} -1+y1 & 1+y2 & y3 \\ 1 & -1 & 0 \\ 0 & -1 & -1 \end{bmatrix} \quad \textbf{(1.1.2.7)}$$

```
> piRp:=det(N)=0;
```

$$piRp := y1 + y2 - y3 = 0 \quad \textbf{(1.1.2.8)}$$

c)

```
> r[eq1]:=2*x1+x2=0;
  r[eq2]:=x1-2*x2+x3=1;
```

$$r_{eq1} := 2\,x1 + x2 = 0$$
$$r_{eq2} := x1 - 2\,x2 + x3 = 1 \quad \textbf{(1.1.3.1)}$$

```
> solve({r[eq1],r[eq2]},{x1,x2,x3});
```

$$\{x1 = x1, x2 = -2\,x1, x3 = -5\,x1 + 1\} \quad \textbf{(1.1.3.2)}$$

```
> par_r:=[a,-2*a,1-5*a];
```

$$par\_r := [a, -2\,a, 1 - 5\,a] \quad \textbf{(1.1.3.3)}$$

The parametric equations of r in R' are:

```
> RRp(a, -2*a, 1-5*a);
```

$$[1 + a, 1 + 3\,a, -1 + 5\,a] \quad \textbf{(1.1.3.4)}$$

The coordinates in R' of a point T of r and the coordinates in B' of a vector u in the direction of r:

```
> TRp:=[1,1,-1];uBp:=[1,3,5];
```

$$TRp := [1, 1, -1]$$
$$uBp := [1, 3, 5] \quad \textbf{(1.1.3.5)}$$

Continuous equations:

```
> (x1-1)/1=(x2-1)/3;(x2-1)/3=(x3+1)/5;
```

$$x1 - 1 = \frac{1}{3}\,x2 - \frac{1}{3}$$
$$\frac{1}{3}\,x2 - \frac{1}{3} = \frac{1}{5}\,x3 + \frac{1}{5} \quad \textbf{(1.1.3.6)}$$

```
> P:=matrix([Y-TRp,uBp]);
```

$$P := \begin{bmatrix} -1+y1 & -1+y2 & 1+y3 \\ 1 & 3 & 5 \end{bmatrix} \quad \textbf{(1.1.3.7)}$$

```
> rRp1:=det(submatrix(P,1..2,[1,2]))=0;rRp2:=det(submatrix
  (P,1..2,[1,3]))=0;
```

$$rRp1 := -2 + 3\,y1 - y2 = 0$$
$$rRp2 := -6 + 5\,y1 - y3 = 0 \quad \textbf{(1.1.3.8)}$$

(Execise13, sheet 4)

r =(1,-2,3)+<(1,-3,2)>; s = (-2,0,1)+<(4,-5,4)>

```
> restart;with(linalg):
```

a) Check that the intersection of r and s is not empty.

b) Find the parametic and cartesian equations of the plane generated by r and s.